
==========================NPN_GUMMEL==========================

HOW TO DO MODEL VERIFICATION FOR CURRENT,VOLTAGE,AND BETA. 

NPN_GUMMEL
* MEASURE IS,ISE,NF,NE,RE,IKF,BETA 
*
*                 ________
*                |        | VC
*                |C      _|_
*         B     _|      /2v \
*         ____|'  npnv  \___/
*        |    |`->        |
*    VB _|_      |  0     |
*      /.7v\     |________|
*      \___/      _|_
*        |        ///
*       _|_
*       ///

.OPTIONS  GMIN=1e-15    METHOD=gear   ABSTOL=1e-15 
*======== ====== ====== ====== ====== ====== ====== ====== ======
VC        C      0      DC     5V 
VB        B      0      0V 
Q1        C      B      0      NPNV 

.MODEL    NPNV   NPN(
*========================================================== 
+IS=1.1E-18      NF=1.005   BF=220        VAF=130      IKF=5e-03 
+ISE=9.15E-15    NE=2 
*==========================================================                                          
+CJE=2E-12 CJC=2E-13 CJS=3E-12 TF=.6E-9 )

.control 
*DC       SOURC1 VSTART VSTOP  VSTEP  SOURC2 START2 STOP2  STEP2
dc        vb     .4     1.2    .1    
set       pensize = 2 
plot      mag(-i(vc)) mag(-i(vb)) vs mag(V(b))      ylog   title Gummel 
plot      mag(i(vc)/i(vb))        vs mag(vc#branch) loglog title Beta_vs_IC 
.endc 
.end 

**                                ___
**                               |(C)|    NPN
**     SPICE MODEL               |___|
**    <actual_behavior>            |
**    <varies from spice model !>  / RC 110ohms    VAF=215
**                                 \
**                                 /                    4fF
**         ________________________|_____________________
**        |       |      |      |    |      |     _|_   _|_Cjs
**        |       |12fF  |      |    |Ir/BR |Irn / _ \  ___      
**        |       |      |      /   _|_    _|_   \/ \/   |gnd!
**       _|_Cjcx _|_Cjc _|_Cdc  \    ^      ^    /\_/\  _|_
**       ___     ___    ___     /   /_\    /_\   \___/ \sub/
**   ___  |   Rbb'|      | gmin \    |      |   Ic |    \ /
**  |(B)|_|_/\  __|______|______|____|______|      | |   V
**  |___|     \/  |      |      |   _|_    _|_     | V (hidden)
**      600ohms  _|_Cje _|_Cde  \   \ /    \ /     |
**     BF=116    ___    ___     /   _v_    _v_     |
**                |20fF  | gmin \    |If/Bf |Ifn   |
**                |______|______|____|______|______|
**                                 |
**                  TF=8ps         / RE 45ohms
**                                 \
**                                 /
**                                _|_           
**                               |(E)|
**                               |___|
**            
**


